



Χωρική Απεικόνιση Αστικού Ιστού

Τρισδιάστατη Αυτοματοποιημένη
Σχεδίαση Όγκου Δεδομένης Κάτοψης

Τεχνική έκθεση ανάπτυξης εφαρμογής για το μάθημα
ΤΕΧΝΙΚΕΣ ΠΡΟΓΡΑΜΜΑΤΙΣΜΟΥ ΚΑΙ ΧΡΗΣΗ Η/Υ ΣΤΙΣ
ΚΑΤΑΣΚΕΥΕΣ

Γάκιος Βασίλειος ΑΕΜ:13097

Περιεχόμενα

Περιεχόμενα.....	2
1. Το πραγματικό πρόβλημα	3
2. Συμβασεις.....	3
3. Περιγραφή Εφαρμογής	3
3.1 Διάταξη και επαναληπτική διαδικασία.....	5
3.2 Εισαγωγή Δεδομένων.....	6
3.3 Αποθήκευση αρχείου (.txt)	10
3.4 Άνοιγμα αρχείου (.txt).....	12
3.5 Υπορουτίνες.....	13
3.6 Σχεδιασμός	15
3.6.1 Καθορισμός υψών περιοχών	15
3.6.2 Άνοιγμα AutoCAD.....	15
3.6.3 Δημιουργία layer για κάθε περιοχή.....	15
3.6.4 Επαναληπτική διαδικασία σχεδιασμού	15
3.6.5 Προσαρμογή οπτικής γωνίας	17
4 Παραδείγματα εφαρμογής	18
4.1 kukliko.txt (2 περιοχές)	18
4.2 kukliko.txt (4 περιοχές)	19
4.3 kukliko.txt (8 περιοχές)	20
4.4 serres.txt.....	21

1. Το πραγματικό πρόβλημα

Στο πλαίσιο του μαθήματος ΠΕΡΙΦΕΡΕΙΑΚΗ ΑΝΑΠΤΥΞΗ ΙΙ. ΘΕΩΡΙΕΣ ΚΑΙ ΕΦΑΡΜΟΓΕΣ ΠΡΟΓΡΑΜΜΑΤΙΣΜΟΥ-ΣΧΕΔΙΑΣΜΟΥ και πιο συγκεκριμένα σε θέμα με τίτλο Η ΕΞΕΛΙΞΗ ΤΟΥ ΠΟΛΕΟΔΟΜΙΚΟΥ ΣΧΕΔΙΑΣΜΟΥ ΚΑΙ ΤΩΝ ΟΡΩΝ ΔΟΜΗΣΗΣ ΣΤΗΝ ΠΟΛΗ ΤΩΝ ΣΕΡΡΩΝ παρουσιάστηκε η ανάγκη δημιουργίας ενός δυναμικού εργαλείου απεικόνισης των αλλαγών που επιφέρουν οι εκάστοτε αλλαγές στις τιμές των όρων δόμησης. Η ανάγκη για απλοποίηση και αστικοποίηση των αριθμητικών δεδομένων που συλλέχτηκαν στην συγκεκριμένη βιβλιογραφική ερευνά ,αλλά και σε κάθε έρευνα, έχει ως σκοπό την ευκολότερη κατανόηση των αποτελεσμάτων από άτομα μη σχετικά με το αντικείμενο και άρα την αύξηση της εξωστρέφειας του. Βασικός σκοπός ,λοιπόν, της εφαρμογής είναι να αποτελέσει εργαλείο απεικόνισης της σχέσης ορών δόμησης-όγκου αστικού ιστού.

2.Συμβασεις

Επειδή το πραγματικό πρόβλημα της ακριβής διαμόρφωσης της γεωμετρίας του αστικού ιστού συναρτήσει των όρων δόμησης περιγράφεται από πάρα πολλές παραμέτρους (συντελεστής κάλυψης ,συντελεστής δόμησης, συντελεστής δ κ.α.) καθίσταται αδύνατος ο ακριβής σχεδιασμός του. Για αυτό το λόγο είναι απαραίτητο να γίνουν κάποιες συμβάσεις οι οποίες είναι:


- αγνόηση όλων των συντελεστών πλην του συντελεστή δόμησης
- αντικατάσταση του συντελεστή δόμησης με το ύψος των όγκων της κάθε περιοχής

Με λίγα λόγια, η εφαρμογή σαν είσοδο δέχεται τα απαραίτητα στοιχεία της κάτοψης (γεωμετρία, ταξινόμηση σε περιοχές) και τα ύψη που αντιστοιχούν στην κάθε περιοχή ανάλογα με το συντελεστή δόμησης της και δημιουργεί για κάθε οικόπεδο ένα στερεό (Acad3DSolid) με βάση το εμβαδό του οικοπέδου και ύψος αυτό της περιοχής στην οποία ανήκει.


3.Περιγραφή Εφαρμογής

Η εφαρμογή αναπτύχθηκε στο περιβάλλον της VBA του Microsoft Excel και μέσω του ActiveX δίνει τις αντίστοιχες εντολές στην VBA του Autocad για σχεδιασμό. Ανοίγοντας


το αρχείο sxediasmos όπως φαίνεται στην Εικόνα 1 ανοίγει το αρχείο που φαίνεται στην εικόνα 2. Πατώντας στο button "Άνοιγμα φόρμας" εμφανίζεται η φόρμα που φαίνεται στην εικόνα 3.

Name	Date modified	Type	Size
 sxediasmos	11/2/2016 5:39 μμ	Φύλλο εργασίας τ...	38 KB

Εικόνα 1: Αρχείο προγράμματος



Εικόνα 2: Απεικόνιση προγράμματος αμέσως μετά το άνοιγμα



Εικόνα 3: Η βασική φόρμα

Όπως φαίνεται στην Εικόνα 3 η φόρμα αποτελείται από 4 buttons η ανάλυση των οποίων γίνεται στην συνέχεια. Πριν γίνει αυτό είναι απαραίτητο να παρουσιαστεί ο τρόπος

διάταξης των δεδομένων στο φύλλο εργασίας του Microsoft Excel και η επαναληπτική διαδικασία με την οποία επιτυγχάνεται αυτή.

3.1 Διάταξη και επαναληπτική διαδικασία

Για λόγους ευκρίνειας των δεδομένων αλλά και ευκολίας στην επεξεργασία τους κρίνεται σκόπιμο η διάταξη τους να ακολουθεί κάποιους κανόνες. Η μορφή που επιλέχθηκε φαίνεται στην Εικόνα 4.

Αριθμός Οικοπέδων				
Αριθμός Περιοχών				
Κόμβοι Οικοπέδου 1	Τετμημένη Κόμβου 1	Τετμημένη Κόμβου 2	Τετμημένη Κόμβου n1
Περιοχή Οικοπέδου 1	Τεταγμένη Κόμβου 1	Τεταγμένη Κόμβου 2	Τεταγμένη Κόμβου n1
Κόμβοι Οικοπέδου 2	Τετμημένη Κόμβου 1	Τετμημένη Κόμβου 2	Τετμημένη Κόμβου n2
Περιοχή Οικοπέδου 2	Τεταγμένη Κόμβου 1	Τεταγμένη Κόμβου 2	Τεταγμένη Κόμβου n2
.....
.....
Κόμβοι Οικοπέδου n	Τετμημένη Κόμβου 1	Τετμημένη Κόμβου 2	Τετμημένη Κόμβου n
Περιοχή Οικοπέδου n	Τεταγμένη Κόμβου 1	Τεταγμένη Κόμβου 2	Τεταγμένη Κόμβου n

Εικόνα 4: Διάταξη δεδομένων στο φύλλο εργασίας του Excel

Η παραπάνω διάταξη χωρίζεται σε ζώνες των τριών σειρών (δύο σειρές δεδομένων και μια κενή σειρά για το διαχωρισμό των δεδομένων) στον κατακόρυφο άξονα, ενώ στον οριζόντιο άξονα το μήκος της κάθε σειράς εξαρτάται από τον αριθμό των κόμβων του κάθε οικοπέδου και διαφέρει για κάθε ζώνη των τριών σειρών. Λόγω αυτής της ιδιαιτερότητας χρειάζεται να οριστούν:

- ένας κατακόρυφος (καθολικός) δείκτης i
- και ένας οριζόντιος (τοπικός) δείκτης j

Δείκτης i : Ο δείκτης αυτός δείχνει σε ποιά από τις ζώνες βρίσκεται η επανάληψη. Επειδή όμως η πρώτη ζώνη δεν αναφέρεται σε κάποιο οικόπεδο ορίζεται και ο δείκτης $n=i+1$. Έτσι ο αριθμός κόμβων του πρώτου οικοπέδου ($i=1$) βρίσκεται στο Cells($3*n-2,1$) για $n=2$.

Δείκτης j : Ο δείκτης αυτός δείχνει σε ποιά κόμβο βρίσκεται η επανάληψη. Ονομάζεται τοπικός γιατί δεν έχουν όλα τα οικόπεδα των ίδιο αριθμό κόμβων

Οπότε η επαναληπτική διαδικασία έχει ως εξής:

Cells(1,1)= arithmos_oikopedwn

Cells(2,1)=arithmos_perioxwn

For i=1 to Cells(1,1)

n=i+1

Cells(3*n-2,1)=arithmos_komvwn

Cells(3*n-1,1)=arithmos_perioxhs

for j=1 to Cells(3*n-2,1)

Cells(3 * n - 2, j + 1)=tetmhmenh_oikopedou_i_komvou_j


Cells(3 * n - 1, j + 1)= tetagmenh_oikopedou_i_komvou_j



Next




Next

3.2 Εισαγωγή Δεδομένων

Ακολουθώντας την παραπάνω επαναληπτική διαδικασία και με τη χρήση των InputBoxes ζητούνται τα δεδομένα από το χρήστη με την προβολή κατάλληλων κατατοπιστικών μηνυμάτων. Παρουσιάζονται ενδεικτικά κάποιες εικόνες παρακάτω.








```

Private Sub CommandButton1_Click()

'koumpi 1 eisagwgh dedomenwn

On Error Resume Next 'se periptwsh lanthsmenhs eisagwghs den stamataei h leitourgia toy programmatos

N_oik = InputBox("Δώστε το πλήθος των οικοπέδων", "Εισαγωγή Στοιχείων", "Πλήθος οικοπέδων") 'prwta eisagwgonται ta katholika dedomena etsi wste na arithmi
Cells(1, 1) = N_oik
N_reg = InputBox("Δώστε το πλήθος των περιοχών", "Εισαγωγή Στοιχείων", "Πλήθος περιοχών")
Cells(2, 1) = N_reg

If N_oik < N_reg Then
MsgBox ("Δεν γίνεται οι περιοχές να είναι περισσότερες από τα οικόπεδα")
End If

For i = 1 To N_oik 'prwth epanalhps (katakorfos aksonas)
'epeidh sthn thesh toy prwtou oikopedoy exoun mpei ta katholika dedomena prepei na orisoume to deikth n
n = i + 1
Cells(3 * n + -2, 1) = InputBox("Δώστε τον αριθμό των κόμβων του οικοπέδου " & i, "Εισαγωγή Στοιχείων", "αριθμός των κόμβων του οικοπέδου" & i)
'molis eisaxthei o arithmos kombwn orizetai kai to megethos toy orizontioy(topikoy) aksona
Cells(3 * n - 1, 1) = InputBox("Δώστε τον αριθμό της περιοχής του οικοπέδου " & i, "Εισαγωγή Στοιχείων", "αριθμός της περιοχής του οικοπέδου" & i)

For j = 1 To Cells(3 * n + -2, 1) 'deuterh epanalhps (orizontios aksonas). arithmos epanalhpsewn=arithmos kombwn
Cells(3 * n - 2, j + 1) = InputBox("Δώστε την τετμημένη του κόμβου " & j & " του οικοπέδου " & i, "Εισαγωγή Στοιχείων", "X" & i & j)
Cells(3 * n - 1, j + 1) = InputBox("Δώστε την τεταγμένη του κόμβου " & j & " του οικοπέδου " & i, "Εισαγωγή Στοιχείων", "Y" & i & j)
Next
Next

elegxosdedomenwn = 1 'orismos deikth elegxou dedomenwn


End Sub

```

Εικόνα 5: Κώδικας "Εισαγωγή Δεδομένων"

3.3 Αποθήκευση αρχείου (.txt)

Για λόγους φορητότητας των δεδομένων δίνεται η δυνατότητα αποθήκευσης των δεδομένων σε αρχείο κειμένου. Τα δεδομένα αποθηκεύονται σε μορφή διανύσματος όπως φαίνεται στην Εικόνα 5



Εικόνα 6: Μορφή αρχείου κειμένου

Αυτό επιτυγχάνεται με την ίδια επαναληπτική διαδικασία, αφού πρώτα γραφτούν οι απαραίτητες εντολές για την "ενεργοποίηση" του αρχείου και με τη χρήση της εντολής `arχειο.WriteLine Cells(i,j)` πραγματοποιείται η μεταφορά των δεδομένων στο αρχείο. Ο κώδικας βρίσκεται στην Εικόνα 7. Ο ορισμός του ονόματος και της "τοποθεσίας" (η διαδρομή του αρχείου π.χ. C:\Program Files\AutoCAD 2010) γίνονται με τη χρήση του παραθύρου "Save as" των windows.

```

Private Sub CommandButton3_Click()
Dim deikths As Integer

If elegxosdedomenwn = 0 And Cells(1, 1) = 0 Then 'elegxos yparkshs dedomenwn

    MsgBox "Πρέπει να εισάγετε δεδομένα"

Else
    Dim fso As Scripting.FileSystemObject
    Dim ts As Scripting.TextStream
    Set fso = New Scripting.FileSystemObject
    Dim varResult As Variant

    varResult = Application.GetSaveAsFilename(FileFilter:="text files (*.txt), *.txt") 'anoigma formas apothikeysh arxeiow
    Set ts = fso.CreateTextFile(varResult) 'dhmioyrgia arxeiow (*.txt)
    'ksekinai h "grafh" twm dedomenon sto arxeiow keimenoy. ola ta dedomena apothikeuontai san ena dianusma
    ts.WriteLine Cells(1, 1)
    ts.WriteLine Cells(2, 1)

    For i = 1 To Cells(1, 1) 'akoloutheitai h idia epanalhpthikh diadikasia kai gia kathe ena keli apotheuetai mia grammh sto arxeio
        n = i + 1
        ts.WriteLine Cells(3 * n - 2, 1)
        ts.WriteLine Cells(3 * n - 1, 1)

        For j = 1 To Cells(3 * n - 2, 1)
            ts.WriteLine Cells(3 * n - 2, j + 1)
            ts.WriteLine Cells(3 * n - 1, j + 1)
        Next

    Next

    ts.Close
    Set fso = Nothing
End If

End Sub

```

Εικόνα 7: Κώδικας "Αποθήκευση αρχείου"

3.4 Άνοιγμα αρχείου (.txt)

Όπως φαίνεται και στην εικόνα 8, γίνεται αρχικά ένας έλεγχος για το αν επιλέχθηκε κάποιο αρχείο στο παράθυρο "Open file" των windows και αν ναι τότε αποθηκεύει στα κελιά τις αντίστοιχες τιμές με την ίδια επαναληπτική διαδικασία.

```
Private Sub CommandButton2_Click()  
    'koumpi 2 anoigma arxeiou  
    Dim doxeio As Double 'metablhth gia na exasfalistei oti auto poy tha apothike  
    Dim intChoice As Integer  
    Dim strPath As String  
    Dim fso As Scripting.FileSystemObject  
    Dim ts As Scripting.TextStream  
    Dim onoma As Variant  
  
    Application.FileDialog(msoFileDialogOpen).AllowMultiSelect = False 'apotrepei t  
    intChoice = Application.FileDialog(msoFileDialogOpen).show 'anoigei to parathu  
  
    If intChoice <> 0 Then 'logikos elegchos gia to an exei ginei epilogh arxeiou  
  
        strPath = Application.FileDialog(msoFileDialogOpen).SelectedItem(1)  
        onoma = strPath 'apothikeush toy path toy arxeiou  
        Set fso = New Scripting.FileSystemObject  
        Set ts = fso.OpenTextFile(onoma) 'anoigma arxeiou me onoma to path tou|  
  
        'enarksh diadikasia eisagwghs dedomenwn apo arxeio (*.txt). ta dedomena s  
        N_oik = ts.ReadLine  
        Cells(1, 1) = N_oik  
        N_reg = ts.ReadLine  
        Cells(2, 1) = N_reg  
  
        If N_oik < N_reg Then  
            MsgBox ("Δεν γίνεται οι περιοχές να είναι περισσότερες από τα οικόπεδα")  
        Else  
  
            'akoloutheitai h idia epalanhptikh diadikasia me thn eisagwgh dedomenwn.  
            For i = 1 To N_oik  
                n = i + 1  
                doxeio = ts.ReadLine  
                Cells(3 * n - 2, 1) = doxeio  
  
                doxeio = ts.ReadLine  
                Cells(3 * n - 1, 1) = doxeio  
  
                For j = 1 To Cells(3 * n - 2, 1)  
                    doxeio = ts.ReadLine  
                    Cells(3 * n - 2, j + 1) = doxeio  
                    doxeio = ts.ReadLine  
                    Cells(3 * n - 1, j + 1) = doxeio  
                Next  
            Next  
  
            elegxosdedomenwn = 1  
            ts.Close  
            Set fso = Nothing  
        End If  
    End If  
End Sub
```

Εικόνα 8: Κώδικας "Άνοιγμα αρχείου"

3.5 Υπορουτίνες

Στις παρακάτω εικόνες παρουσιάζονται κάποιες υπορουτίνες που εξυπηρετούν την αυτοματοποιημένη συγγραφή κώδικα.

```
'υπορουτινα που ανοιγει το autocad
Sub AcadConnect()

On Error Resume Next

Set AcadApp = GetObject(, "Autocad.Application")
If Err Then
    Err.Clear
    Set AcadApp = CreateObject("Autocad.Application")
    AcadApp.Visible = True
    If Err Then
        MsgBox Err.Description
        Exit Sub
    End If
End If

End Sub
```

Εικόνα 9: Υπορουτίνα για άνοιγμα AutoCAD

```
'υπορουτινα που prosthetei neo layer
Sub AcadNewLayer(LayerName As String, LayerClr%)
    Set LayerObj = AcadApp.ActiveDocument.Layers.Add(LayerName)
    LayerObj.Color = LayerClr%
End Sub
```

Εικόνα 10: Υπορουτίνα που προσθέτει νέο layer

```
'υπορουτινα που kleinei to autocad

Sub AcadClose()
    Set AcadApp = Nothing
End Sub
```

Εικόνα 11: Υπορουτίνα που κλείνει το autocad

```

'υπορουτινα gia grammi
Sub AcadLine(xs As Double, ys As Double, xe As Double, ye As Double, LayerName As String)
    Dim sp(0 To 2) As Double
    Dim ep(0 To 2) As Double

    sp(0) = xs
    sp(1) = ys
    sp(2) = 0#
    ep(0) = xe
    ep(1) = ye
    ep(2) = 0#

    Set LineObj = AcadApp.ActiveDocument.ModelSpace.AddLine(sp, ep)

    LineObj.Layer = LayerName
End Sub

```

Εικόνα 12: Υπορουτίνα για σχεδιασμό γραμμής

Οι παραπάνω υπορουτίνες έχουν ληφθεί από το αρχείο του Microsoft Excel "γρولوجιμος_S-_kamrylhs" που δόθηκε στο μάθημα ΟΔΟΠΟΪΙΑ Ι, ο δημιουργός του οποίου απεικονίζεται στην Εικόνα 13

```

'.....
' H BASIKH DOMI APO TO RAPORT EINAI TOY
'G.F.Vartalitis 2009
'.....
'ΟΔΟΠΟΪΙΑ Ι
'By FIDELIO 2009
'-----

```

Εικόνα 13: Στοιχεία δημιουργού πηγής

3.6 Σχεδιασμός

3.6.1 Καθορισμός υψών περιοχών

Το πρώτο δεδομένο που είναι απαραίτητο για τον σχεδιασμό του όγκου είναι τα ύψη των περιοχών. Με την χρήση μιας επανάληψης και ενός διανύσματος αποθηκεύονται τα ύψη των περιοχών (Εικόνα 14).

```
Dim yprsh(100) As Double 'orismos dianusmatos ypawn perioxwn  
For i = 1 To Cells(2, 1) 'epanalhptikh diadikasia gia oles tis perioxes Cells(2,1)=plithos_perioxwn  
    On Error Resume Next  
    yprsh(i) = InputBox("Δώστε το ύψος των κτιρίων της περιοχής " & i, "Εισαγωγή Στοιχείων", "Υψος περιοχής " & i)  
Next
```

Εικόνα 14

3.6.2 Άνοιγμα AutoCAD

```
AcadConnect 'anoigma autocad
```

Εικόνα 15

3.6.3 Δημιουργία layer για κάθε περιοχή

Δημιουργούνται όσα layer όσες και οι περιοχές.

```
For i = 1 To Cells(2, 1) 'dhmiourgia diaforetikoy  
    AcadNewLayer "perioxh" & i, 3 * i  
Next
```

Εικόνα 16

3.6.4 Επαναληπτική διαδικασία σχεδιασμού

Παρουσιάζεται η διαδικασία για τυχαίο οικόπεδο. Για να δημιουργηθεί το ζητούμενο στερεό απαιτείται σαν είσοδος μια περιοχή (region) και ένα ύψος(height). Για να δημιουργηθεί μια περιοχή(region) απαιτείται σαν είσοδος ένα διάνυσμα (curves). Τα περιεχόμενα του διανύσματος είναι οι καμπύλες που συνθέτουν την περίμετρο της περιοχής. Άρα ο κώδικας για το τυχαίο οικόπεδο οφείλει να περιέχει τις παραπάνω κατευθύνσεις.

1. Αρχικά αποθηκεύονται ο αριθμός των κόμβων του οικοπέδου και ο αριθμός της περιοχής και διαστασιολογείται το διάνυσμα. Το διάνυσμα διαστασιολογείται κάθε φορά για κάθε οικόπεδο (ReDim) διότι είναι εργαλείο του οριζόντιου (τοπικού άξονα). (Εικόνα 17)

```
komvoi = Cells(3 * n - 2, 1)
perioxh = Cells(3 * n - 1, 1)
ReDim curves(komvoi - 1) As AcadEntity
```

2. Το επόμενο βήμα είναι η χάραξη και η αποθήκευση των γραμμών στο διάνυσμα (Εικόνα 18).

```
For j = 1 To komvoi
    'dhmioyrgountai lines metaksu twv kombwn
    If j = komvoi Then 'eisagwgh elegxoy etsi wste na xaraxthei kai h grammh metaksu prwtoy kai teleutaiou komvou
        AcadLine Cells(3 * n - 2, j + 1), Cells(3 * n - 1, j + 1), Cells(3 * n - 2, 2), Cells(3 * n - 1, 2), "perioxh" & perioxh
        Set curves(j - 1) = LineObj
    Else
        AcadLine Cells(3 * n - 2, j + 1), Cells(3 * n - 1, j + 1), Cells(3 * n - 2, j + 2), Cells(3 * n - 1, j + 2), "perioxh" & perioxh
        Set curves(j - 1) = LineObj
    End If
Next
```

3. Μετά γίνεται η δημιουργία της περιοχής

```
Dim regionObj As Variant
regionObj = AcadApp.ActiveDocument.ModelSpace.AddRegion(curves)
```


4. Ορίζονται οι τιμές των ορισμάτων και δημιουργείται το στερεό

```
' Define the extrusion
Dim height As Double
Dim taperAngle As Double
height = ypsh(perioxh)
taperAngle = 0

' Create the solid
Dim solidObj As Acad3DSolid 'dhmiourgia tou stereou
Set solidObj = AcadApp.ActiveDocument.ModelSpace.AddExtrudedSolid(regionObj(0), height, taperAngle)
solidObj.Layer = "perioxh" & perioxh
```

3.6.5 Προσαρμογή οπτικής γωνίας

Τέλος γίνεται προσαρμογή της οπτικής γωνίας με τη χρήση του διανύσματος κατεύθυνσης.


```
Dim NewDirection(0 To 2) As Double
NewDirection(0) = -1: NewDirection(1) = -1: NewDirection(2) = 1
```

Αφού οριστεί το διάνυσμα, ορίζεται και η οπτική γωνία με βάση αυτό και στο τέλος ZoomAll.


```
AcadApp.ActiveDocument.ActiveViewport.Direction = NewDirection
AcadApp.ActiveDocument.ActiveViewport = AcadApp.ActiveDocument.ActiveViewport
ZoomAll
```

4 Παραδείγματα εφαρμογής


4.1 kukliko.txt (2 περιοχές)



4.2 kukliko.txt (4 περιοχές)



4.3 kukliko.txt (8 περιοχές)



4.4 serres.txt

